Representations of Curves and Surfaces, and of their Tangent Lines, and Tangent Planes in R^2 and R^3

Robert L.Foote, Fall 2007

Multivariable calculus is inherently geometric. Calculus gives us tools to define and study smooth curves and surfaces. Curves and surfaces can be represented in three ways, as graphs, as level sets, and parametrically. Some can be represented in all three of these ways; others can be represented in only one or two ways. The way a curve or surface is represented determines how you work with it. In particular, it determines how you compute things associated with it, such as tangent lines or planes, normal vectors, and approximations.

Each section contains a linear example. The curves and surfaces associated with linear functions are lines and planes. Linear functions are easy to work with algebraically and the associated lines and planes are easy to visualize geometrically. Moreover, tangent lines and planes to more complicated curves and surfaces serve as models for how the more complicated objects behave. Thus, linear functions are important special cases that can serve to motivate more general situations.

Note that the linear examples given here are linear in the high school sense, but not the Math 223 sense. (A function f is linear in the Math 223 sense if $f(x+y) = f(x) + f(y)$ and $f(a x) = a f(x)$ for all x and y in the domain of f and all numbers a. Generally x and y are vectors, but they can be numbers also. The function $f(x) = m x$ is linear in the Math 223 sense, but the function $f(x) = m x + b$ is not. The technical term for a function that is linear in the high school sense but not the Math 223 sense is *affine*.)

Note: There are several graphics you can manipulate with sliders in this notebook. You may need to reevaluate them to get them to work.

Note: There is a pdf version of this, which you can print. It is not meant to be a substitute for working with the *Mathematica* notebook, however. I suggest you print it and use it to take notes on as you go through the notebook.

Curves and Surfaces

Graphs

- **The graph of $f : \mathbb{R} \to \mathbb{R}$ is a curve in $\mathbb{R}^2$**
- **The graph of $f : \mathbb{R}^2 \to \mathbb{R}$ is a surface in $\mathbb{R}^3$**
- **The graph of $f : \mathbb{R}^k \to \mathbb{R}^n$ is a k-dimensional surface in $\mathbb{R}^{k+n}$**
- **The graph of $f : \mathbb{R} \to \mathbb{R}^2$ is a curve in $\mathbb{R}^3$**

Level Sets; In Particular, Level Curves and Level Surfaces

A level set of a function $f : \mathbb{R}^k \to \mathbb{R}^n$ is what you get in the domain of f when you set the output of the function equal to a constant.

- **A level set of $f : \mathbb{R}^2 \to \mathbb{R}$ is typically a curve in $\mathbb{R}^2$**
- **Why level sets are "level"**
- **A level set of $f : \mathbb{R}^3 \to \mathbb{R}$ is typically a surface in $\mathbb{R}^3$**
- **A graph is always a level set. A level set is usually a graph, at least locally.
Not finished.**
- **A level set of $f : \mathbb{R}^3 \to \mathbb{R}^2$ is typically a curve in $\mathbb{R}^3$
Not started.**

Parametric Curves and Surfaces

- **The image of $\gamma : \mathbb{R} \to \mathbb{R}^n$ is a curve in $\mathbb{R}^n$**
  - **Parameterized lines
Add P and v to the first picture**
  - **Non-linear examples**

- **The image of $F:\mathbb{R}^2 \to \mathbb{R}^3$ is a surface in $\mathbb{R}^3$**
  - **Parameterized planes**
  - **Torus example**
- **A graph can always be parameterized.   A parameterized set is usually a graph, at least locally.**
  Not started.

Representations of Curves and Surfaces, Summary

Tangent Lines and Planes, and Linear Approximations
Not started.

Representations of Curves and Surfaces, and of their Tangent Lines, and Tangent Planes in R^2 and R^3

Robert L.Foote, Fall 2007

Multivariable calculus is inherently geometric. Calculus gives us tools to define and study smooth curves and surfaces. Curves and surfaces can be represented in three ways, as graphs, as level sets, and parametrically. Some can be represented in all three of these ways; others can be represented in only one or two ways. The way a curve or surface is represented determines how you work with it. In particular, it determines how you compute things associated with it, such as tangent lines or planes, normal vectors, and approximations.

Each section contains a linear example. The curves and surfaces associated with linear functions are lines and planes. Linear functions are easy to work with algebraically and the associated lines and planes are easy to visualize geometrically. Moreover, tangent lines and planes to more complicated curves and surfaces serve as models for how the more complicated objects behave. Thus, linear functions are important special cases that can serve to motivate more general situations.

Note that the linear examples given here are linear in the high school sense, but not the Math 223 sense. (A function f is linear in the Math 223 sense if $f(x+y) = f(x) + f(y)$ and $f(a\,x) = a\,f(x)$ for all x and y in the domain of f and all numbers a. Generally x and y are vectors, but they can be numbers also. The function $f(x) = m\,x$ is linear in the Math 223 sense, but the function $f(x) = m\,x + b$ is not. The technical term for a function that is linear in the high school sense but not the Math 223 sense is *affine*.)

Note: There are several graphics you can manipulate with sliders in this notebook. You may need to reevaluate them to get them to work.

Curves and Surfaces

Graphs

- **The graph of $f : \mathbb{R} \to \mathbb{R}$ is a curve in $\mathbb{R}^2$**

This is familiar from Math 111 and 112. Here is the graph of $f(x) = \frac{1}{2}x - 1$. Note that any non-vertical line is the graph of some linear function.

```
f[x_] = 1/2 x - 1;
Plot[f[x], {x, -1, 4}]
```



Here is the graph of $f(x) = \sin x$.

```
Plot[Sin[x], {x, -Pi, Pi}]
```



Here is the graph of a more "generic" function. (Don't worry about what an interpolating polynomial is.)

```
f[x_] = InterpolatingPolynomial[{{0, 1.5}, {2, 2}, {4, 1.5}, {5, 2}}, x];
Plot[f[x], {x, -.5, 5}, PlotRange → {0, 2.5}]
```



■ The graph of $f : \mathbb{R}^2 \to \mathbb{R}$ is a surface in $\mathbb{R}^3$

This is familiar from the multivariable part of Math 112. The graph of a linear function $f : \mathbb{R}^2 \to \mathbb{R}$ is a plane, and any non-vertical plane is the graph of such a function.

```
f[x_, y_] = 3 x - 2 y + 3;
Plot3D[f[x, y], {x, -3, 3}, {y, -3, 3}]
```



Here is the graph of $f(x, y) = \sin x + \cos y$. You can rotate the graph with the mouse. Try it!

```
Plot3D[Sin[x] + Cos[y], {x, -Pi, Pi}, {y, -Pi, Pi}]
```



Here is the graph of $f(x, y) = e^{-x^2 - y^2}$, the two-variable version of the bell curve.


```
f[x_, y_] = Exp[-x^2 - y^2];
Plot3D[f[x, y], {x, -2, 2}, {y, -2, 2}, PlotRange → All]
```



■ The graph of $f : \mathbb{R}^k \to \mathbb{R}^n$ is a k-dimensional surface in $\mathbb{R}^{k+n}$

For example, the graph of $f : \mathbb{R}^3 \to \mathbb{R}$ is a 3-dimensional surface in $\mathbb{R}^4$, and the graph of $f : \mathbb{R}^2 \to \mathbb{R}^3$ is a 2-dimensional surface in $\mathbb{R}^5$. For the *graph* of a function, every dimension of the domain and range must be taken into account. When $k + n > 3$, we can't visualize the graph, but that doesn't stop us from talking about it. We will work with functions of the type $f : \mathbb{R}^2 \to \mathbb{R}^3$ and $f : \mathbb{R}^3 \to \mathbb{R}$, but we will visualize them parametrically and with their level surfaces, respectively.

■ The graph of $f : \mathbb{R} \to \mathbb{R}^2$ is a curve in $\mathbb{R}^3$

We generally don't use this type of graph because it is easier to deal with curves in more than two dimensions parametrically. In fact, *Mathematica* doesn't have a way to directly plot the graph of such a function. Instead, it has to be plotted parametrically. Here is the graph of $f(x) = (\cos x, \sin 3x)$. You want to think of this as $(y, z) = f(x) = (\cos x, \sin 3x)$, that is, y and z are functions of x. The x-axis runs from left to right in the picture. Planes perpendicular to the x-axis intersect the curve in at most one point, which is the version of the "vertical line test" in this context. To get the feeling of the shape of the curve, you need to rotate the picture. A particularly interesting view is had by pointing the x-axis directly towards you. This gives you the *image* of f as a parametric curve in $\mathbb{R}^2$. (See parametric curves and surfaces below.)

```
ParametricPlot3D[{x, Cos[x], Sin[3 x]}, {x, 0, 2 Pi}, AspectRatio → Automatic]
```



Linear example. Here is a linear example that is familiar from Math 223. Consider the equations

$$2x - y + z = 4 \qquad \text{and} \qquad -4x + y + z = 2.$$

Geometrically, these are non-parallel planes in $\mathbb{R}^3$, and so they intersect in a line. In linear algebra you learned how to solve them as a system. Since there are two equations in three unknowns, you expect there to be one free (independent) variable. If we choose x to be the independent variable and solve for y and z in terms of x, we get

$$y = 3x - 1 \qquad \text{and} \qquad z = x + 3.$$

You can think of y and z as separate functions of x, or you can think of the point (y, z) as a function of x, namely, $(y, z) = f(x) = (3x - 1, x + 3)$. In linear algebra you probably let $x = t$ and wrote the solutions as $(x, y, z) = X(t) = (t, 3t - 1, t + 3)$, which emphasizes the parametric viewpoint. The *graph* of f and the *image* of X are the same line in $\mathbb{R}^3$.

```
ParametricPlot3D[{x, 3 x - 1, x + 3}, {x, -1, 1}]
```



Level Sets; In Particular, Level Curves and Level Surfaces

A level set of a function $f : \mathbb{R}^k \to \mathbb{R}^n$ is what you get in the domain of f when you set the output of the function equal to a constant.

■ A level set of $f : \mathbb{R}^2 \to \mathbb{R}$ is typically a curve in $\mathbb{R}^2$

Let $z = f(x, y) = 4x^2 + 9y^2$. The level set corresponding to $z = 36$ is the curve with equation $4x^2 + 9y^2 = 36$, or $\frac{x^2}{9} + \frac{y^2}{4} = 1$, which is an ellipse. Here is the graph. Note: This is the graph of the *equation* $4x^2 + 9y^2 = 36$. It is *not* the graph of the function f. The graph of f is a *surface* in $\mathbb{R}^3$, as seen in the section on graphs.


```
f[x_, y_] = 4 x^2 + 9 y^2;
ContourPlot[f[x, y], {x, -4, 4}, {y, -2.5, 2.5},
 Contours → {36}, AspectRatio → Automatic, ContourShading → False]
```



Here are several level curves of this function.

```
ContourPlot[f[x, y], {x, -4, 4}, {y, -2.5, 2.5}, AspectRatio → Automatic, ContourShading → False]
```



A level set of $f : \mathbb{R}^2 \to \mathbb{R}$ is a *typically* a curve, but not always. For $z = f(x, y) = 4x^2 + 9y^2$ there are some values of $z = c$ for which you don't get a curve. The equation $4x^2 + 9y^2 = c$ has no solutions when $c < 0$. Thus, the level set for $z = c < 0$ is the empty set. The equation $4x^2 + 9y^2 = 0$ has only one solution, namely $(0, 0)$. Thus, the level set for $z = 0$ consists of just one point. Note that the level curves are not equally-spaced in the domain (the *xy*-plane), but their *z*-values *are* equally-spaced. (You can see what the *z*-value is for a level curve by hovering the mouse cursor over a point of the curve.) This indicates that the function increases more quickly the further you get from the origin.

Here is a more interesting example. The level curve that crosses itself is for the constant output $z = 0$. This particular curve is called the Folium of Descartes.

```
g[x_, y_] = x^3 + y^3 - 3 x y;
ContourPlot[g[x, y], {x, -3, 3}, {y, -3, 3},
 Contours → 20, AspectRatio → Automatic, ContourShading → False]
```



Note again that the level curves are not equally-spaced. The function changes more quickly as you cut across the level curves in the places where the curves are spaced more densely.

A linear function $f:\mathbb{R}^2 \to \mathbb{R}$ has the form $f(x, y) = a\,x + b\,y + c$. Setting this to the constant d yields $a\,x + b\,y = C$, where $C = d - c$, which is the equation of a line. You can get any line in this way, including vertical lines. Note that the level curves are equally-spaced, which indicates that a linear function increases at a constant rate as you cut across the level curves.

```
f[x_, y_] = 3 x - 2 y + 5;
ContourPlot[f[x, y], {x, -4, 4}, {y, -4, 4}, ContourShading → False]
```



■ Why level sets are "level"

Let $g(x, y) = x^3 + y^3 - 3\,x\,y$ again. Below is its graph being sliced by some horizontal plane $z = c$, and the level curve corresponding to that value. You can use the slider to change the value of c. The horizontal slice through the graph is called a *trace*. A trace is congruent to the corresponding level curve, but they are not the same. The trace is in $\mathbb{R}^3$, where the graph of f is, whereas the level curve is in the domain, $\mathbb{R}^2$. The term "level" simply comes from the horizontal slices being level.

Note that functions $f:\mathbb{R}^2 \to \mathbb{R}$ are the *only* functions for which it is reasonable to visualize *both* the graph and the level sets. The idea of horizontally slicing the graph gives an important connection between the graph and the level curves.

```
In[1]:= g[x_, y_] = x^3 + y^3 - 3 x y;
        Manipulate[GraphicsArray[{Plot3D[{g[x, y], c}, {x, -3, 3}, {y, -3, 3}, PlotRange → {-5, 10.2}],
           ContourPlot[g[x, y], {x, -3, 3}, {y, -3, 3}, Contours → {c}, PlotPoints → 20,
            AspectRatio → Automatic, ContourShading → False]}], {{c, 0}, -5, 10, Appearance → "Open"}]
```

Out[2]=



The picture would get crowded, but you can easily imagine slicing the graph with multiple equally-spaced horizontal planes. The corresponding level curves are *not* equally-spaced -- they are more densely spaced corresponding to the places where the graph is steeper.

The situation for a linear example is simpler, of course.

In[3]:=
```
f[x_, y_] = 3 x + 2 y + 5;
Manipulate[GraphicsArray[{Plot3D[{f[x, y], c}, {x, -3, 3}, {y, -3, 3}],
   ContourPlot[f[x, y], {x, -3, 3}, {y, -3, 3}, Contours → {c}, PlotPoints → 20,
    AspectRatio → Automatic, ContourShading → False]}], {{c, 0}, -5, 10, Appearance → "Open"}]
```

Out[4]=



▪ A level set of $f:\mathbb{R}^3 \to \mathbb{R}$ is typically a surface in $\mathbb{R}^3$

The level sets of a linear function $f:\mathbb{R}^3 \to \mathbb{R}$ are equally-spaced planes. Rotate the picture to get the full effect. The planar regions are not congruent because of the way they are being cut off by the bounding box.

```
g[x_, y_, z_] = x + 2 y + 3 z - 5;
ContourPlot3D[g[x, y, z], {x, -3, 3}, {y, -3, 3}, {z, -3, 3}]
```



Since multiple surfaces are hard to visualize together, level surfaces are usually drawn one at a time. Here are three level surfaces of $w = f(x, y, z) = x^2 + y^2 - z^2$ corresponding to $w = 1$, $w = 0$, and $w = -1$, plotted separately. Note that the first and last are

nice surfaces, but the middle one has an unusual point.

```
f[x_, y_, z_] = x^2 + y^2 - z^2;
ContourPlot3D[f[x, y, z], {x, -3, 3}, {y, -3, 3}, {z, -3, 3}, Contours → {1}]
```



```
ContourPlot3D[f[x, y, z], {x, -3, 3}, {y, -3, 3}, {z, -3, 3}, Contours → {0}]
```



```
ContourPlot3D[f[x, y, z], {x, -3, 3}, {y, -3, 3}, {z, -3, 3}, Contours → {-1}]
```



In[5]:=
```
f[x_, y_, z_] = x^2 + y^2 - z^2;
Manipulate[ContourPlot3D[f[x, y, z], {x, -3, 3}, {y, -3, 3}, {z, -3, 3}, Contours → {c}],
 {c, -1, 1, Appearance → "Open"}]
```

Out[6]=



A graph is always a level set. A level set is usually a graph, at least locally.
Not finished.

This example shows that a level set isn't always a nice curve. Another example is the level set of $z = f(x, y) = 4x^2 + 9y^2$ for the constant output $z = 0$. The only solution of $4x^2 + 9y^2 = 0$ is (0, 0), and so this level set is a point, not a curve.

- **A level set of $f : \mathbb{R}^3 \to \mathbb{R}^2$ is typically a curve in $\mathbb{R}^3$**
  Not started.

Parametric Curves and Surfaces

- **The image of $\gamma : \mathbb{R} \to \mathbb{R}^n$ is a curve in $\mathbb{R}^n$**

  - **Parameterized lines**
    Add P and v to the first picture

The linear case is a parameterized line, $X = \gamma(t) = P + t\,\mathbf{v}$, or in coordinates, $x = x_0 + a\,t$, $y = y_0 + b\,t$. Note that the line you see is *not the graph* of γ, rather it is the *image* of γ. You can get any line in this way.

In[11]:=
```
P = {2, 3};  v = {2, -1};
c[t_] = P + t v
line = ParametricPlot[P + t v, {t, -2, 2}, AxesOrigin → {0, 0}]
```

Out[12]= {2 + 2 t, 3 - t}

Out[13]=



You really want to think about this as a point moving along the line as a function of time, that is, you need to see how the point $c(t)$ depends on t. The point P and vector $\mathbf{v}$ are also shown.

In[47]:=
```
P = {2, 3};  v = {2, -1};
vector = Arrow[{P, P + v}];
line = ParametricPlot[P + t v, {t, -2, 2}, AxesOrigin → {0, 0}];
Manipulate[Show[line, Graphics[{PointSize[.03], Point[P], Point[P + t v], vector}]],
 {t, -2, 2, Appearance → "Open"}]
```

Out[50]=



- **Non-linear examples**

Here is a familiar example, the standard parameterization of an ellipse.

```
c[t_] = {3 Cos[t], 2 Sin[t]};
ellipse = ParametricPlot[c[t], {t, 0, 2 Pi}]
```



Here is the point moving along the ellipse.

In[51]:=
```
c[t_] = {3 Cos[t], 2 Sin[t]};
ellipse = ParametricPlot[c[t], {t, 0, 2 Pi}]; Manipulate[
 Show[ellipse, Graphics[{PointSize[.03], Point[c[t]]}]], {t, 0, 2 Pi, Appearance → "Open"}]
```

Out[52]=



Here is another curve.

In[20]:=
```
c[t_] = {3 Sin[2 t], 2 Cos[t]};
image = ParametricPlot[c[t], {t, 0, 2 Pi}];
Manipulate[Show[image, Graphics[{PointSize[.03], Point[c[t]]}]],
 {t, 0, 2 Pi, Appearance → "Open"}]
```

Out[22]=



The image of $F : \mathbb{R}^2 \to \mathbb{R}^3$ is a surface in $\mathbb{R}^3$

▪ Parameterized planes

The linear case is a parameterized plane, $X = F(s, t) = P + s\,\mathbf{v} + t\,\mathbf{w}$, where **v** and **w** are independent vectors. Note that you can get any plane in this way. Here is an example. The point *P* and the vectors **v** and **w** are also shown in the picture.

```
P = {3, 2, 5};  v = {1, 3, 2};  w = {3, 2, -1};
F[s_, t_] = P + s v + t w
vSegment = {Thick, Red, Line[{P, P + v}]};
wSegment = {Thick, Yellow, Line[{P, P + w}]};
plane = Show[{ParametricPlot3D[F[s, t], {s, -2, 2}, {t, -2, 2}],
   Graphics3D[{vSegment, wSegment, PointSize[.03], Point[P]}]}]
```

```
{3 + s + 3 t, 2 + 3 s + 2 t, 5 + 2 s - t}
```



The image, the picture of the plane, only conveys part of the information. To visualize the mapping, you need to see how the point on the plane, $F(s, t)$, moves in $\mathbb{R}^3$ as (s, t) moves in $\mathbb{R}^2$.

Here are two ways to control *s* and *t*.


```
Manipulate[ListPlot[{{s, t}}, PlotRange → {{-2.1, 2.1}, {-2.1, 2.1}},
  PlotStyle → {PointSize[.03], Red}, AxesLabel → {"s", "t"}],
 {s, -2, 2, Appearance → "Open"}, {t, -2, 2, Appearance → "Open"}]
```



In the next method, use the mouse to move the orange dot on the left. You can think of the square with the orange dot as being part of the (s, t) plane in the obvious way.

```
Manipulate[ListPlot[{sandt}, PlotRange → {{-2.1, 2.1}, {-2.1, 2.1}},
  PlotStyle → {PointSize[.03], Red}, AxesLabel → {"s", "t"}],
 {{sandt, {0, 0}, "(s,t)"}, {-2, -2}, {2, 2}}, ControlType → Slider2D, ControlPlacement → Left]
```



Now here is the point $F(s, t)$ being controlled by (s, t) in these two ways. Remember that you can use the mouse to rotate the image.

In[53]:=
```
P = {3, 2, 5};  v = {1, 3, 2};  w = {3, 2, -1};  F[s_, t_] = P + s v + t w;
vSegment = {Thick, Red, Line[{P, P + v}]};  wSegment = {Thick, Yellow, Line[{P, P + w}]};
plane = Show[{ParametricPlot3D[F[s, t], {s, -2, 2}, {t, -2, 2}],
   Graphics3D[{vSegment, wSegment, PointSize[.03], Point[P]}]}]; Manipulate[
 Show[plane, Graphics3D[{PointSize[.03], Point[P], Point[F[s, t]], vSegment, wSegment}]],
 {s, -2, 2, Appearance → "Open"}, {t, -2, 2, Appearance → "Open"}]
```

Out[54]=



In[28]:=
```
P = {3, 2, 5};  v = {1, 3, 2};  w = {3, 2, -1};
F[{s_, t_}] = P + s v + t w;
vSegment = {Thick, Red, Line[{P, P + v}]};  wSegment = {Thick, Yellow, Line[{P, P + w}]};
plane = Show[{ParametricPlot3D[F[s, t], {s, -2, 2}, {t, -2, 2}],
   Graphics3D[{vSegment, wSegment, PointSize[.03], Point[P]}]}]; Manipulate[
 Show[plane, Graphics3D[{PointSize[.03], Point[P], Point[F[sandt]], vSegment, wSegment}]],
 {{sandt, {0, 0}, "(s,t)"}, {-2, -2}, {2, 2}}, ControlType → Slider2D, ControlPlacement → Left]
```

Out[30]=



- **Torus example**

Here is a map, Torus: $\mathbb{R}^2 \to \mathbb{R}^3$, the image of which is a torus. The map takes two angles, θ and φ, and uses them to compute a point in $\mathbb{R}^3$. Thinking of (θ, φ) as a point in $\mathbb{R}^2$, we get a map from two-dimensional space into three-dimensional space.

In[33]:=
```
U[θ_] = {Cos[θ], Sin[θ], 0};
V[θ_, φ_] = Cos[φ] U[θ] + Sin[φ] {0, 0, 1};
Torus[θ_, φ_] = 3 U[θ] + V[θ, φ]
torus = ParametricPlot3D[Torus[θ, φ], {θ, 0, 2 Pi}, {φ, 0, 2 Pi}]
```

Out[35]=
```
{3 Cos[θ] + Cos[θ] Cos[φ], 3 Sin[θ] + Cos[φ] Sin[θ], Sin[φ]}
```

Out[36]=



To visualize the mapping, you need to see how the point on the torus, Torus(θ, φ), moves in $\mathbb{R}^3$ as (θ, φ) moves in $\mathbb{R}^2$. The entire torus is covered for θ and φ in the ranges $0 \le \theta \le 2\pi$ and $0 \le \varphi \le 2\pi$. Here is a point on the torus controlled by θ and φ in two ways. Remember that you can use the mouse to rotate the picture.

In[37]:=
```
U[θ_] = {Cos[θ], Sin[θ], 0}; V[θ_, φ_] = Cos[φ] U[θ] + Sin[φ] {0, 0, 1};
Torus[θ_, φ_] = 3 U[θ] + V[θ, φ];
torus = ParametricPlot3D[Torus[θ, φ], {θ, 0, 2 Pi}, {φ, 0, 2 Pi}];
Manipulate[Show[torus, Graphics3D[{PointSize[.03], Red, Point[Torus[θ, φ]]}]],
 {{θ, 5. Pi / 3}, 0, 2 Pi, Appearance → "Open"},
 {{φ, Pi / 3.}, 0, 2 Pi, Appearance → "Open"}, ControlType → {Automatic, "Open"}]
```

Out[38]=



In[42]:=
```
U[θ_] = {Cos[θ], Sin[θ], 0};  V[θ_, φ_] = Cos[φ] U[θ] + Sin[φ] {0, 0, 1};
Torus[{θ_, φ_}] = 3 U[θ] + V[θ, φ];
torus = ParametricPlot3D[Torus[θ, φ], {θ, 0, 2 Pi}, {φ, 0, 2 Pi}];
Manipulate[Show[torus, Graphics3D[{PointSize[.03], Red, Point[Torus[θandφ]]}]],
 {{θandφ, {5. Pi / 3, Pi / 3.}, "(θ,φ)"}, {0, 0}, {2 Pi, 2 Pi}},
 ControlType → Slider2D, ControlPlacement → Left]
```

Out[44]=



In the next example, a portion of the image is drawn for $0 \le \theta \le \theta_{max}$ and $0 \le \varphi \le \varphi_{max}$, and the sliders control θ_{max} and φ_{max}.

In[45]:=
```
U[θ_] = {Cos[θ], Sin[θ], 0};  V[θ_, φ_] = Cos[φ] U[θ] + Sin[φ] {0, 0, 1};
Torus[θ_, φ_] = 3 U[θ] + V[θ, φ];
torus = ParametricPlot3D[Torus[θ, φ], {θ, 0, 2 Pi}, {φ, 0, 2 Pi}];
Manipulate[ParametricPlot3D[Torus[θ, φ], {θ, 0, θmax}, {φ, 0, φmax},
  PlotRange → {{-4, 4}, {-4, 4}, {-1, 1}}], {{θmax, Pi / 2., "θmax"}, .1, 2 Pi, Appearance → "Open"},
 {{φmax, Pi / 2., "φmax"}, .1, 2 Pi, Appearance → "Open"}]
```

Out[46]=



- **A graph can always be parameterized.   A parameterized set is usually a graph, at least locally.**
  Not started.

Representations of Curves and Surfaces, Summary

- A graph represents full information about a function, and requires dimensions for *both the domain and range* of the function
- Level sets of a function are viewed in the *domain* of the function. The values of the function (in the range) are sometimes represented by labeling the level sets. The only functions for which we can visualize both the graph and level curves are $f : \mathbb{R}^2 \to \mathbb{R}$, and the relationship between these two viewpoints is important in understanding both.
- A function that is a parameterization is visualized by its image, which is in the *range* of the function. For parameterized curves, the input (domain) values are sometimes represented by labeling selected points along the curve. The only functions

for which we can visualize both the graph and the parametric image are $\gamma : \mathbb{R} \to \mathbb{R}^2$, however the parametric image is almost always the preferred viewpoint.

Tangent Lines and Planes, and Linear Approximations
Not started.